Behavioral model of Parallel Optical Modules

Dr Yannick HERVE[1], Dr Patricia DESGREYS[2]

CNRS-PHASE[1]
Pôle API, Bd S.BRANT
67400 ILLKIRCH, FRANCE
herve@erm1.u-strasbg.fr

ENST[2]
Laboratoire Traitement et Communication de l'Information (LTCI)
46 rue Barrault, 75634 PARIS Cedex 13
patricia.desgreys@enst.fr

Abstract

VHDL-AMS allows designers to use any level of abstraction to model objects (1). We present a model of a parallel optical module in accordance with the *Optical Internetworking Forum* (OIF) specifications (14). Users can use it to plan performances of optical links between boards or cabinets. This approach allows the virtual prototyping of global systems.

Introduction

To increase the amount of exchanged data, designers can use new products like POM (Parallel Optical Modules)(10). These objects, organized around a VCSEL array (1D or 2D), can exchange data from 8Gb/s to 480 Gb/s. It is very important that this modules share the same specifications in order to ensure interoperability.

The *Optical Internetworking Forum* (OIF) is a worldwide organization which promote standard and recommendation. As it can be read on their web site (14) : "The mission of the Optical Internetworking Forum (OIF) is to foster the development and deployment of interoperable products and services for data switching and routing using optical networking technologies. The OIF will encourage co-operation among telecom industry participants including equipment manufacturers, telecom service providers and end users; promote global development of optical internetworking products; promote nationwide and worldwide compatibility and interoperability; encourage input to appropriate national and international standards bodies; and identify, select, and augment as appropriate and publish optical internetworking specifications drawn from appropriate national and international standards."

In order to validate the architecture of a network, it is very important to be able to simulate at a system level and detailed models are unsuitable for this task.(2-5) and ("Opto-electro-thermal Model of a VCSEL array using VHDL-AMS" in the same conference).

This paper describes high level models of Very Short Range Optical modules. The efficiency of simulations due to the simplicity of the models allows to validate architectures at a system level. The main objective of this work is to use VHDL-AMS to produce light models of complex systems but also to show that VHDL-AMS is usable to write formal and executable specifications.

Specifications

The Interconnect Optical Forum (OIF) proposes specifications for emitters and receivers. The VSR (Very Short Range) modules fall into 4 categories. Each of them are specified through a set of parameters. Table 1 and Table 2 show the differences between families and specifications for the VSR-1 emitter.

Using VHDL-AMS

VHDL was designed to support many tasks in the design process of digital integrated circuit and systems (11-13).Its extension, VHDL-AMS, allows to design analog parts of the systems. It is reputed not to support specification level. Nevertheless its high level of semantic allows to write models at very high level of abstraction (1)(8)(9).

If a generic parameter is used directly to drive the value of an output, the model is considered as a formal and executable specification. In this reported work, each parameter of the specification is used as a generic parameter. The simulation results are directly driven by the values.

The ideal VCSEL model

The model is built by instanciating the device on the eight ideal paths. The ability of the language to describe structural view is plenty used.

The interface for each path, the ideal VCSEL, of the model is organized around generic parameters, an input signal and an optical terminal. In this example we choose to describe the optical terminal by an electrical one.

```
library disciplines;
use disciplines.electromagnetic_system.all;
entity Vcsel_beh is
 generic(skew : time := 5000 ps;
         jitter : time := 50 ps;
         P_1 : real := 0.1;
         Text : real := 4.0;
         Tr : real := 260.0e-12;
         Bp : real := 1.5e9;
         P_noise : real := 0.025);
 port(signal data : in bit;
      terminal T_Po : electrical);
end;
```

	Length	Fiber type	Fiber number	Laser type	Wavelength
VSR-1	300 m	MMF	12	VCSEL	850 nm
VSR-2	600 m	SMF	1	FP	1310 nm
VSR-3	300 m	MMF	4	VCSEL	850 nm
VSR-4	300 m	MMF	1	VCSEL	850 nm

MMF : Multi Mode Fiber, SMF : Single Mode Fiber, VCSEL : Vertical Cavity / FP : Fabry-Perrot

Table 1 : Four families of VSR module

	Min	Max	Units
Bit rate	1.244160 +/- 20ppm		Gb/s
Optical Power	-10	-9	dBm
Wavelength	830	860	nm
Extinction rate	6		dB
Wavelength variation		0.85	nm
Rise and fall time		260	ps
RIN		-117	DB/Hz
Skew		5	ns
Jitter		+/- 50	ps

Table 2 : Specifications of VSR-1 emitter module

This ideal VCSEL is described at a high level of abstraction. The model hides current generator, bias current, threshold and thermal modulation effects, ...

```
library ieee;
use ieee.math_real.all;
architecture VSR_1_BEH of Vcsel_beh is
 signal Q_skew,Q_jitt, clock : bit;
 signal Q_ideal, Q_noise : real := 0.0;
 Quantity Q_noised: real;
 Quantity Po across Io through T_Po;
begin
    clock <= not clock after 100 ps;
    -- Noise generation
    process
      variable unf : real;
      variable seed1:integer:=7648;
      variable seed2:integer:=45;
     begin
       wait on clock;
       Uniform(seed1,seed2,unf);
       Q_noise <= (unf-0.5) * P_noise;
    end process;
    -- Skew genration
    Q_skew <= transport data after skew-jitter/2;
    -- random jitter generation
    process
      variable unf : real;
      variable seed1:integer := 3456;
      variable seed2:integer := 4563;
     begin
      wait on Q_skew;
      Uniform(seed1,seed2,unf);
      Q_jitt <= transport Q_skew after unf*jitter;
    end process;
    -- Simulation kernels synchronisation
    break on Q_jitt,Q_noise;
    -- Analog quantity affectation
    process
     begin
      wait on Q_jitt;
      if Q_jitt='1' then Q_ideal <= P_1;
      else               Q_ideal <= P_1/Text;
      end if;
     end process;
    -- Rising/falling time computation
    Q_noised == Q_ideal'ramp(Tr*10.0/6.0)+Q_noise;
    -- First order filtering / BW
    Po + Po'dot/(2.0*math_pi*bp) == Q_noised;
end;
```

The bit type input signal is directly processed for skew and jitter. Then after this preprocessing, it is associated to a quantity. This quantity is modified by rising/falling time, noise and bandwidth.

The NOISE statement is not supported by the tool we use (Mentor Graphics ADV-MS 1.4.1). Then we chose to generate it by a random local quantity. In order to drive correctly the simulator we must manage the noise by a clock driven process.

The validation process of the ideal VCSEL model is done by a testbench. We input a random digital signal under OIF specification and we draw eye-diagram of the output. This eye diagram is the worst case design of a actual way of a VSR-1 parallel optical module.

```
entity tb_vcsel_beh is
end;

use work.all;
library disciplines;
use disciplines.electromagnetic_system.all;
library ieee; use ieee.math_real.all;
architecture first of tb_vcsel_beh is
 terminal T_p : electrical;
 quantity Po across T_p;
 signal data:bit;
begin
  vcsel:entity Vcsel_beh(VSR_1_BEH)
         port map(data,t_p);

  -- input random signal
  process
    variable seed1 : integer := 2345;
    variable seed2 : integer := 4367;
    variable unf   : real;
    begin
     wait for 800 ps;
     uniform(seed1,seed2,unf);
     if unf > 0.5 then data <= '1';
     else              data <= '0';
     end if;
  end process;
end;
```



Fig 1. :Eye-diagram of the `Vcsel_beh(VSR_1_BEH)` model

We can compare this result to actual measures on a THALES VCSEL (Fig.2.)



Fig 2. : Actual eye-diagram of a THALES VCSEL

The POM model

The complete POM model is built on N instanciations of the ideal VCSEL.

```
entity TRT_POM is
       port (Signal Com(1 to 8):in bit;
             Quantity Po(1 to 8):out real);
end of TRT_POM

Architecture VSR_1_BEH of TRT_POM is
Begin
       for i in 1 to 8 generate
        uut : entity Vcsel_beh(VSR_1_BEH)
              port map (com(i),Po(i));
       end generate;
end;
```

The simulation of the POM gives the same result than the one way model.

Conclusion

This work shows that VHDL-AMS is suitable to describe objects or systems at specification level. These kind of models allows to manage worst case design in virtual prototyping at this level. The designer will be able to predict the BER at this level by association of many objects like emitters, fibers, receivers, ... Using generic parameters is very powerful. The designer can reuse models for other specification without extra job.

These approach will become even more efficient when tools will support all the 1076.1 standard (NOISE, SPECTRUM, FREQUENCY, PROCEDURAL...).Designers are very impatient !

A structural model of interconnected disciplines at behavioral level is studied in the SHAMAN project (PHASE-CNRS, SUPAERO, SUPELEC, ENST, IPSIS, THALES-TRT) funded by the French Minister of Research. This model will take into account electrical, thermal, mechanical, optical aspects of the design in order to optimize the next POM generation.

References

[1] Y. Hervé, "VHDL-AMS: Applications et enjeux industriels" Dunod éditeur, Collection Sciences-sup, ISBN : 2-10-005888-6

[2] P.O. Lauritzen and C.L. Ma, "A simple diode model with reverse recovery", *IEEE Trans. Power Electronics*, vol. 6, pp 188-191, April 1991.

[3] D.E. Dodds, "Fabry-Perrot Laser Diode Modeling", *IEEE Photonics Tech. letters*, vol. 7, n° 3, Mars. 1995.

[4] P.V. Mena, "Circuit-level Modeling and Simulation of Semiconductor Lasers", *Doctoral Thesis*, Department of Electrical and Computer Engineering, University of Illinois, Urbana-Champaign, 1998.

[5] R. Stevens, "Modulation Properties of Vertical Cavity Light Emitters", *Doctoral Thesis*, Department of Microelectronics and Information Technology, Royal Institute of Technology, Stockholm, 2001.

[6] J. M. Bergé, Oz Levia, J. Rouillard,, "Modeling in analog design," Kluwer, ISBN 0-7923-9569-7.

[7] S. A. Ohr, "CAE: A survey of standards, trends and tools", Wiley, ISBN : 0-471-6336-6.

[8] E. Christen et al, "Tutorial VHDL AMS", 36th Design Automation Conference, New Orleans, june 21 25, 1999.

[9] "1076.1-1999 IEEE Standard VHDL Analog and Mixed-Signal Extensions", 320 pages, ISBN : 0-7381-1640-8

[10] "International Technology Roadmap for Semiconductors, Semiconductor Industry Association Edition " , 1999

[11] J-M. Bergé, V. Olive, J. Rouillard, "VHDL : Du langage à la modélisation, CNET/ENST", Presses Polytechniques et Universitaires Romandes, ISBN : 2 88074 361 3.

[12] K. Shakill, "VHDL for programmable logic", Addisson Wesley, ISBN : 0 201 89573 0.

[13] J. WEBER, M. MEAUDRE, "Le Langage VHDL", Dunod, ISBN : 2-10-004755-8.

[14] www.oiforum.com